
This document is based on a one-day short-course lecture on dynamic general equilibrium analysis with
MPSGE. I begin by going through the logic of the Ramsey model which is often presented as a dynamic
optimization problem:

max
∞∑
t=0

(
1

1 + ρ

)t
C1−θ
t − 1
1− θ

s.t.
Ct = f(Kt)− It

Kt+1 = (1− δ)Kt + It

K0 = K̄0

The maximand in this problem is often called constant-elasticity-of-intertemporal substitution (CEIS) utility
function. As will be shown below, it simply represents a montonic transformation of conventional CES utility
function.

Here, as in many macroeconomics textbooks, aggregate output is expressed as a function of the capital stock
alone, i.e.:

Yt = f(Kt)

In the MPSGE representation of the Ramsey model, it is convenient to work with a constant-returns pro-
duction function in which we have inputs of both labor and capital:

Yt = F (L̄t,Kt)

When labor is in fixed supply, the production function exhibits diminishing returns to capital. There is
therefore no loss of generality by formulating the model on the basis of a constant returns to scale technology.

In writing down a model it is helpful to employ the unit cost function associated with the production function
F ():1

c(pLt , r
K
t ) ≡ min pLt aL + rKt aK

s.t.
F (aL, aK) = 1

1Note that the lower-case function c() represents unt cost, while the upper case Ct represents consumption in year t. In the
equilibrium model Ct(p,M) represents the demand for output in year t as a function of output prices and aggregate present
value of income.

Shephard’s lemma tells us that the compensated demand functions for labor and capital are the partial
derivatives of the unit cost function:

aK(rK , pL) =
∂c(pLt , rKt )

∂rKt
and

aL(rK , pL) =
∂c(pLt , rKt )

∂pLt

The representative agent model can be formulated as a general equilibrium model which is completely routine,
apart from the fact that there are an infinite-number of variables. Following the conventional GAMS/MPSGE
framework, equilibrium in the model is characterized by three classes of equations:

1. Market clearance conditions and associated market prices are as follows:2

• Output market (market price pt):
Yt = Ct(p,M) + It

• Labor market (wage rate pLt ):
L̄t = aL(rKt , p

L
t ) Yt

• Market for capital services (capital rental rate rKt ):

Kt = aK(rKt , p
L
t ) Yt

• Capital stock (capital purchase price pKt ):

Kt+1 = (1− δ)Kt + It

2. Zero profit conditions and associated activities are:3

• Output (Yt):
pt = c(pLt , r

K
t )

• Investment (It ≥ 0):
pt ≥ pKt+1

• Capital stock (Kt):
pKt = rKt + (1− δ)pKt+1

3. Income balance:

M = pK0 K̄0 +
∞∑
t=0

pLt L̄t

Two questions might arise for an MPSGE modeller looking at this equilibrium model. First, the careful
observer might note that the demand functions, Ct(p,M) have not been specified, and because these arise
from CIES preferences so there may be some details to work out. This problem is considerably easier than
the second issue, namely how do we solve an infinite-dimensional system of nonlinear equations. Let’s first
consider the issue of CEIS preferences.

Consider the following equivalent representations of intertemporal preferences:
2The demand functions employed in this model assure that all prices will be nonzero in equilibrium. There is no formal

need, therefore, to associated prices with market clearance conditions, as would be required in a conventional complementarity
problem. I provide an associated here in order to help understand how the model might be extended with demand functions
which would admit zero prices.

3The only activity level which could possibly fall to zero would be investment, and that would only happen in a policy
scenario which resulted in a substantial reduction in the return to capital.

1. Additively separable utility:

U(c) =
∞∑
t=0

(
1

1 + ρ

)t
C1−θ
t − 1
1− θ

2. Linearly homogeneous utility:

Û(c) =

[ ∞∑
t=0

(
1

1 + ρ

)t
C1−θ
t

]1/1−θ

It is possible to determine the equivalence of U and Û by recalling that a monotonic transformation of utility
does not alter the underlying preference ordering. Observe that:

Û = V (U) = [aU + κ]1/a

where

κ =
∞∑
t=0

(
1

1 + ρ

)t
=

1 + ρ

ρ
,

and
a = 1− θ.

V () is a monotonic transformation (V ′ > 0), hence optimization of U and Û yield identical demand functions.

Alternatively, recall that preference orderings are defined by the marginal rate of substitution. In both of
these models we have:

∂U/∂Ct+1

∂U/∂Ct
=

1
1 + ρ

(
Ct
Ct+1

)1−θ

There are several advantages associated with the use of linearly homogeneous representation. First of all,
these preferences can be represented in MSPGE. Second, the reporting of welfare changes as Hicksian-
equivalent variations is trivial with Û – a 1% change in Û corresponds to a 1% equivalent variation in
income.

CEIS preferences over a finite horizon can be represented in MPSGE as follows:

$prod:C s:sigma

o:PC q:(c0*sum(t,pref(t)*qref(t)))

i:P(t) q:(c0*qref(t)) p:pref(t)

Intertemporal preferences in an MPSGE model are typically based on the following parameters:

• sigma is the intertemporal elasticity of substitution,

• c0 is the base year consumption level, and

• qref(t) = (1+g0)**(ord(t)-1) is the baseline equilibrium index of economic activity, calculated on
the basis of a steady-state growth rate equal to g0.

• pref(t) = (1+r0)**(ord(t)-1) is the baseline present value price path, calculated on the basis of a
steady-state interest rate equal to r0.

The MPSGE representation includes a discount rate which is define implicitly as:

ρ =
1 + r̄

(1 + g)1−θ − 1

In order to solve a finite approximation of the model with a T -period model horizon, we need to decompose
the consumer’s problem. Consider the infinite-horizon problem of the representative agent in Ramsey’s
model:

max
∞∑
t=0

(
1

1 + ρ

)t
u(ct)

s.t. ∞∑
t=0

pcct = pK0 K̄0 +
∞∑
t=0

pLt L̄t

in which u(c) = c1−θ−1
1−θ , and define a value of terminal assets to be:

A∗T =
∞∑

t=T+1

(
pcc
∗
t − pLt L̄t

)

Then consider the equivalent model:

max
T∑
t=0

(
1

1 + ρ

)t
u(ct) +

∞∑

t=T+t

(
1

1 + ρ

)t
u(ct)

s.t.
T∑
t=0

pcct = pK0 K̄0 +
T∑
t=0

pLt L̄t −AT

∞∑

t=T+1

pcct = AT +
∞∑

t=T+1

pLt L̄t

If AT is fixed then this can be posed as two separate optimization problems, one running through time period
T and another for the post-terminal period. When terminal assets are assigned a value of A∗T , corresponding
to the infinite-horizon solution, then the finite horizon model will then produce consumption levels for years
0 through T which are identical to the ∞-horizon model.

How do we find A∗T ?

Terminal assets in the closed economy model are simply equal to the value of the capital stock at the start
of period T + 1. The model running through year T then produces a good approximiation to the consumer
problem when we have a good approximation to the terminal capital stock. The key insight provided by Lau,
Pahlke and Rutherford (2000) is that the state variable KT+1 can be determined as part of the equilibrium
calculation by targeting the associated control variable, IT . In the present model this could be based on any
of the following primal constraints:

• Terminal investment growth rate set equal to the long-run steady-state growth rate:

IT /IT−1 = 1 + g

• Terminal investment growth rate set equal to the growth rate of aggregate output:

IT /IT−1 = YT /YT−1

• Terminal investment growth rate set equal to the growth rate of consumption:

IT /IT−1 = CT /CT−1

State-variable targeting provides a very compact means of determining the terminal capital stock. In models
with multiple consumers living beyond period T , it would be necessary to account for which of these agents
owns the assets. Note that some agents may have negative asset positions at the end of the model –
particularly in overlapping generations models where young households accumulate debt which is repaid in
middle age.

The final detail involved in implementing a dynamic model in MPSGE is calibration. The simplest approach
to set up the model along a steady-state growth rate in which the interest rate (r̄) and growth rate (ḡ) are
given. The first thing to work out is to determine the structure of the benchmark equilibrium.

Here are the steps involved in sorting out the steady-state conditions which related investment and capital
earnings in a static dataset which is consistent with a steady-state growth path:

1. The zero-profit condition for It reveals the price level for capital:

pKt+1 =
pKt

1 + r̄
= pt

hence
pKt = (1 + r̄)pt

The base year price of capital is then:
p̄K = 1 + r̄

2. The zero profit condition for Kt determines the price level for rKt :

pKt = rKt + (1− δ)pKt+1

Substituting the values of pKt and pKt+1 reveals that the base year rental price of capital is sufficient to
cover interest plus depreciation:

r̄K = r̄ + δ

3. The main challenge involved in calibrating a dynamic model centers on the reconciliation of base year
capital earnings, investment, the steady-state interest rate and the capital depreciation rate. To see
how this works, consider the market clearance condition for capital in the first period:

K1 = K̄0(1− δ) + Ī = (1 + g)K̄0

This implies that base year investment can be calculated on the basis of growth and depreciation of
the base year capital stock:

Ī = K̄0(g + δ)

Finally, we can use r̄K to determine K̄0 on the basis of the value of capital earnings in the base year,
V K, hence:

Ī = V K
g + δ

r + δ

The problem that arises in applied models is that Ī and V K will not satisfy this relation for arbitrary
values of ḡ, r̄ and δ. Something typically has to be adjusted to match up the dataset with the baseline
growth path.

The GAMS code on the following page presents a GAMS/MPSGE model which has been formulated following
these ideas.

1 GAMS Code Describing a Steady-State Equilibrium

This code reads data describing a base year equilibrium. Investment levels are imputed from the base year
capital stock which is in-turn inferred from the assumed capital value share.

1 $TITLE Ramsey Model -- MPSGE formulation calibrated to base year data

2

3 $ontext

4 Calibrate to the steady-state condition:

5

6 I0 = KD0 * (g + delta) / (r + delta)

7

8 where g=2, delta=7, r=5, so

9

10 I0 = 48 * 9 / 12 = 36

11

12 Y I FD

13 P 100 -36 -64

14 PL -52 52

15 RK -48 48

16 PS 36 -36

17

18 $offtext

19

20 PARAMETER g Growth rate /0.02/,

21 r Interest rate /0.05/,

22 delta Depreciation rate /0.07/,

23 kvs Capital value share /0.48/,

24 y0 Base year output,

25 kd0 Base year rental value of capital,

26 k0 Base year capital stock,

27 i0 Base year investment,

28 c0 Base year consumption,

29 l0 Base year labor input,

30 kstock Base year capital stock multiplier /1/,

31 taxk(t) Capital tax rate in period T,

32 qref(t) Reference quantity path,

33 pref(tt) Reference price path;

34

35 * Use the GAMS ORD (ordinality) and CARD (cardinality)

36 * functions to automate the identification of the first

37 * and last periods of the model horizon:

38

39 t0(t) = yes$(ord(t) eq 1);

40 tl(t) = yes$(ord(t) eq card(t));

41 tterm(tt) = yes$(ord(tt) eq card(tt));

42

43 * Calibrate the model to the baseline growth path:

44

45 y0 = 100;

46 kd0 = kvs * y0;

47 l0 = y0 - kd0;

48 k0 = kd0 / (r + delta);

49 i0 = (g + delta) * k0;

50 c0 = y0 - i0;

51 taxk(t) = 0;

52 qref(t) = (1+g)**(ord(t)-1);

53 pref(tt) = (1/(1+r))**(ord(tt)-1);

54

55 DISPLAY y0, kd0, l0, k0, i0, c0, g, r, delta;

2 GAMS Code for Benchmark Replication of the Model

This section of the code declares the GAMS/MPSGE model, assigns steady-state values for activity levels
and price, and then checks consistency of the resulting model.

1

2 * Define the time horizon of the model using two sets, one which

3 * includes the post-terminal year and one which covers only the

4 * endogenous years:

5

6 SET tt Time horizon (with the first year of the post-terminal period) /2004*2081/,

7 t(tt) Time period over the model horizon /2004*2080/;

8 t0(t) Year 0,

9 tl(t) Last endogenous year,

10 tp(tt) First post-terminal year;

11

12 $ONTEXT

13

14 $MODEL:RAMSEY

15

16 $SECTORS:

17 Y(t) ! Output

18 I(t) ! Investment

19 K(t) ! Capital stock

20

21 $COMMODITIES:

22 P(t) ! Output price

23 RK(t) ! Return to capital

24 PK(tt) ! Capital price

25 PL(t) ! Wage rate

26

27 $CONSUMERS:

28 RA ! Representative agent

29

30 $AUXILIARY:

31 TK ! Post-terminal capital stock

32

33 $PROD:Y(t) s:1

34 O:P(t) Q:Y0

35 I:PL(t) Q:L0

36 I:RK(t) Q:KD0 A:RA T:TaxK(t)

37

38 $PROD:K(tt)$T(tt)

39 O:PK(TT+1) Q:(K0*(1-DELTA))

40 O:RK(tt) Q:KD0

41 I:PK(tt) Q:K0

42

43 $PROD:I(tt)$T(tt)

44 O:PK(TT+1) Q:I0

45 I:P(tt) Q:I0

46

47 $DEMAND:RA s:1

48 D:P(t) Q:(qref(t)*C0) P:pref(t)

49

50 E:PL(t) Q:(L0*qref(t))

51 E:PK(T0) Q:(K0*KSTOCK)

52 E:PK(TP) Q:-1 R:TK

53

54 $REPORT:

55 V:C(t) D:P(t) DEMAND:RA

56 V:W W:RA

57

58 $CONSTRAINT:TK

59 SUM(T$TL(T+1), I(T+1)/I(t) - Y(T+1)/Y(t)) =E= 0;

60

61 $OFFTEXT

62 $SYSINCLUDE mpsgeset RAMSEY

63

64 * Assign steady-state equilibrium values for quantities

65 * and prices:

66

67 Y.L(t) = qref(t);

68 I.L(t) = qref(t);

69 K.L(t) = qref(t);

70

71 P.L(t) = pref(t);

72 RK.L(t) = pref(t);

73 PL.L(t) = pref(t);

74

75 * The steady-state price of capital is the output price

76 * times one plus the interest rate:

77

78 PK.L(tt) = (1+r) * pref(tt);

79 TK.L = k0 * (1+g)**card(t);

80

81 RAMSEY.ITERLIM = 0;

82 $INCLUDE RAMSEY.GEN

83 SOLVE RAMSEY USING MCP;

84 ABORT$(RAMSEY.OBJVAL > 0.001) "Model does not calibrate.";

85 RAMSEY.ITERLIM = 10000;

3 GAMS Code for a Policy Experiment

This code assigns a tax on capital earnings beginning in year 6. The resulting equilibrium is computed
assuming that economic agents anticipate the application of the tax, resulting in a sharp response in invest-
ment and other economic variables to the new economic environment. Over time the tax leads to a reduction
in the steady-state capital stock and the real wage.

Output are presented in graphs using GNUPLOT, both to the windows screen and as encapsulated postscript.

1

2 * Apply a tax on capital inputs of 25% beginning in year 6:

3

4 TAXK(t)$(ORD(t) > 5) = 0.25;

5

6 $INCLUDE RAMSEY.GEN

7 SOLVE RAMSEY USING MCP;

8

9 * Generate some reports with graphs:

10

11 PARAMETER indices "Consumption, Investment and Capital Stock Indices";

12 indices(t,"C") = C.L(t)/(c0*qref(t));

13 indices(t,"I") = I.L(t)/qref(t);

14 indices(t,"K") = K.L(t)/qref(t);

15

16 DISPLAY INDICES;

17

18 * Define the domain over which the X-axis will be defined:

19

20 $setglobal domain t

21

22 * Define the labels to be printed along the X axis:

23

24 set tlbl(t) Time periods to be labelled in output plots /2010,2030,2050,2070/;

25

26 $setglobal labels tlbl

27

28

29 * Place the key to the figures outside the graph (see GNUPLOT 3.7 Help File):

30

31 $setglobal gp_opt0 "set key outside"

32

33 * Plot the graph with horizonal and vertical grid lines (see GNUPLOT 3.7 Help File):

34

35 $setglobal gp_opt1 "set grid"

36

37 * Generate the plot to the screen -- it can subsequently be copied to the clipboard

38 * using a right-click of the mouse, and then pasted into a separate program for

39 * publication:

40

41 $libinclude plot indices

42

43 * Repeat the report generation process a couple more times:

44

45 parameter price "Capital prices and wage rate";

46 price(t,"RK") = RK.L(t)/P.L(t);

47 price(t,"PK") = PK.L(t)/((1+r)*P.L(t));

48 price(t,"PL") = PL.L(t)/P.L(t);

49

50

51 $libinclude plot price

52

53 parameter grates(t,*) Growth rates through the transition;

54

55 grates(t,"c") = 100 * (C.l(t+1)/C.l(t) - 1);

56 grates(t,"i") = 100 * (I.l(t+1)/I.l(t) - 1);

57 grates(t,"k") = 100 * (K.l(t+1)/K.l(t) - 1);

58

59 set gs /c,i,k/;

60 grates(tl,gs) = na;

61 $libinclude plot grates

62

63 * Produce some EPS graphs to put into the notes:

64

65 $setglobal batch yes

66 $setglobal gp_opt2 "set term postscript eps monochrome ’Times-Roman’ 16"

67 $setglobal gp_opt3 "set title"

68 $setglobal gp_opt0 "set key outside width 4"

69

70 $setglobal gp_opt4 "set output ’price.eps’"

71 $libinclude plot price

72 $setglobal gp_opt4 "set output ’indices.eps’"

73 $libinclude plot indices

74 $setglobal gp_opt4 "set output ’grates.eps’"

75 $libinclude plot grates

9



Figure 1: Consumption, Investment and Capital Stock Indices

Figure 2: Capital Prices and Wage Rates

Figure 3: Growth Rates through the Transition

